Volumenerhaltung bei Divergenz = 0

Buch:Höhere Mathematik sehen und verstehen, Haftendorn, Riebesehl, Dammer, Springer Spektrum, Feb. 2021

Datei Vektorfeld-Divergenz-anschaulich-zentral.nb Abschnitt 3.2.5.4 Seite 250, Abb. 3.22

Das Vektorfeld

Dies ist ein Standardfeld mit ∇F=0, es stellt im 2-dimensionalen Raum eine Zentralkraft dar.

```
F = {x/(x^2 + y^2), y/(x^2 + y^2)};
```

Bestimmung von Bahnkurven, die entlang *F* fließen

Vorbereitung für die Differentialgleichung mit DSolve

```
tof = {x → x[t], y → y[t]};
```

Lösen der DGln mit Anfangspunkt (t=0) $\begin{pmatrix} a \\ b \end{pmatrix}$:

```
sol = {x[t], y[t]} /.
   DSolve[{x'[t] == (F[[1]] /. tof), y'[t] == (F[[2]] /. tof), x[0] == a, y[0] == b},
     {x[t], y[t]}, t][[2]] // PowerExpand // Simplify
```

$$\left\{\frac{a\sqrt{a^2+b^2+2t}}{\sqrt{a^2+b^2}}, \frac{b\sqrt{a^2+b^2+2t}}{\sqrt{a^2+b^2}}\right\}$$

Der Anfangspunkt wandert von $\begin{pmatrix} a \\ b \end{pmatrix}$ nach $\begin{pmatrix} c \\ d \end{pmatrix}$, wenn s von 1 nach 0 läuft *), wir erhalten eine Parameterdarstellung der Kurve, die aus der Bewegung der Strecke von $\begin{pmatrix} a \\ b \end{pmatrix}$ nach $\begin{pmatrix} c \\ d \end{pmatrix}$ entlang des Feldes F zum Zeitpunkt t entsteht:

```
solc = sol /. {a → a s + (1 - s) c, b → b s + (1 - s) d} // FullSimplify
```

$$\left\{\frac{(c+a s-c s)\sqrt{(c+a s-c s)^2+(d+b s-d s)^2+2t}}{\sqrt{(c+a s-c s)^2+(d+b s-d s)^2}}, \frac{(d+b s-d s)\sqrt{(c+a s-c s)^2+(d+b s-d s)^2+2t}}{\sqrt{(c+a s-c s)^2+(d+b s-d s)^2}}\right\}$$

Bewegung eines Rechtecks entlang der Flusslinien von *F*

Das Rechteck hat die Ecken (1,1), (2,1), (1,3), (2,3).

Die Tabelle im Show läuft der Reihe nach von (1,3) nach (1,1) - linke Kante, von (2,3) nach (2,1) - rechte Kante, von (2,1) nach (1,1) - untere Kante, von (2,3) nach (1,3) - obere Kante.

Man beachte die ungewohnte Richtung, die an der ungeschickten Verwendung des Parameters s liegt. (siehe auch *) weiter oben!)

Das Bild zeigt das gewanderte Rechteck zu ganzzahligen Zeiten von t=0..10.

```
Show[Sequence @@
  Table[ParametricPlot[solc /. {{a → 1, c → 1, b → 1, d → 3}, {a → 2, c → 2, b → 1, d → 3},
        {a → 1, c → 2, b → 1, d → 1}, {a → 1, c → 2, b → 3, d → 3}} /. t → tt //
     Evaluate, {s, 0, 1}], {tt, 0, 10, 1}], PlotRange → All]
```



Berechnung der Fläche des gewanderten Rechtecks

Hier noch einmal die parametrisierten Kurven der Kanten

params = solc /. {{a → 1, c → 1, b → 1, d → 3}, {a → 2, c → 2, b → 1, d → 3}, {a → 1, c → 2, b → 1, d → 1}, {a → 1, c → 2, b → 3, d → 3}}

$$\left\{\left\{\frac{\sqrt{1+(3-2s)^2+2t}}{\sqrt{1+(3-2s)^2}}, \frac{(3-2s)\sqrt{1+(3-2s)^2+2t}}{\sqrt{1+(3-2s)^2}}\right\}, \left\{\frac{2\sqrt{4+(3-2s)^2+2t}}{\sqrt{4+(3-2s)^2}}, \frac{(3-2s)\sqrt{4+(3-2s)^2+2t}}{\sqrt{4+(3-2s)^2}}\right\}, \left\{\frac{(2-s)\sqrt{1+(2-s)^2+2t}}{\sqrt{1+(2-s)^2}}, \frac{\sqrt{1+(2-s)^2+2t}}{\sqrt{1+(2-s)^2}}\right\}, \left\{\frac{(2-s)\sqrt{9+(2-s)^2+2t}}{\sqrt{9+(2-s)^2}}, \frac{3\sqrt{9+(2-s)^2+2t}}{\sqrt{9+(2-s)^2}}\right\}\right\}$$

Hier wird die Fläche unter jeder Kurve berechnet. Die Formel ist einfach $\int_0^1 y[s] \times x'[s]\, ds$

ints = $\int_0^1$ **#[[2]] × D[#[[1]], s] ds & /@ params**

$$\left\{t\left(\frac{1}{5}-\frac{\pi}{4}+\text{ArcTan}[3]\right), t\left(-\frac{4}{65}+\text{ArcTan}\left[\frac{4}{7}\right]\right), -1+\frac{1}{20}t\,(2+5\pi-20\,\text{ArcTan}[2]), -3+t\left(-\frac{21}{130}+\text{ArcTan}\left[\frac{3}{2}\right]-\text{ArcTan}[3]\right)\right\}$$

Jetzt aufgepasst: wie müssen die Flächen addiert oder subtrahiert werden, damit die Fläche im bewegten Rechteck herauskommt? Man malt am besten ein Bild und trägt ein, welches Integral welche Fläche ist. Alle Flächen haben das falsche Vorzeichen, weil s falsch herum läuft.
Wenn man richtig überlegt hat, erkennt man, das die folgende Rechnung richtig ist:

fläche = ints.{-1, 1, 1, -1}

$$2+t\left(-\frac{4}{65}+\text{ArcTan}\left[\frac{4}{7}\right]\right)+\frac{1}{20}t\,(2+5\pi-20\,\text{ArcTan}[2]) - t\left(-\frac{21}{130}+\text{ArcTan}\left[\frac{3}{2}\right]-\text{ArcTan}[3]\right)-t\left(\frac{1}{5}-\frac{\pi}{4}+\text{ArcTan}[3]\right)$$

Puh, was mag das sein?

fläche // Simplify

$$2+\frac{\pi t}{2}+t\left(\text{ArcTan}\left[\frac{4}{7}\right]-\text{ArcTan}\left[\frac{3}{2}\right]-\text{ArcTan}[2]\right)$$

fläche // FullSimplify

2

Passt perfekt: das ist die Fläche des Ursprungsrechtecks, und zwar für jedes t. Damit ist bestätigt, dass ein Fluss entlang eines divergenzfreien Vektorfeldes Flächen erhält.

Nebenbei ergibt sich $\frac{\pi}{2} = \text{ArcTan}\left[\frac{3}{2}\right] + \text{ArcTan}[2] - \text{ArcTan}\left[\frac{4}{7}\right]$